```
;{CR510}
;Comments added 7-7-03 for Paul Stockton (NB)
*Table 1 Program
  01: 2.0000    Execution Interval (seconds)


1:  Pulse (P3) (Pulse channel looking at sensit pc, and ke.  Pc is 5 minute trigger)
 1: 2        Reps      (looks at pulse channel 1 and 2 on logger, pc & ke)
 2: 1        Pulse Channel 1   (Starting pulse channel)
 3: 0        High Frequency, All Counts
 4: 1        Loc [ sensit_1  ]
 5: .0001    Mult  (required sensit multiplier)
 6: 0        Offset

2:  Batt Voltage (P10)   ( Battery voltage to track battery cycles/ problems)
 1: 3        Loc [ battery   ]

3:  Z=F (P30)   (This is our site number. All labled 33333 until in the field then change them                    to
actual site number)
 1: 33333    F
 2: 0        Exponent of 10
 3: 4        Z Loc [ site      ]

4:  Timer (P26)   (Timer, resets to zero every time pc is detected, starts and ends five minute
            data collection)
 1: 5        Loc [ timer     ]

5:  Z=F (P30)   (Like site number this is changed in the field depending on what sensit is
            there.  We put this in to keep track of what sensit is where.)
 1: 2222     F
 2: 0        Exponent of 10
 3: 6        Z Loc [ snstsnmbr ]

6:  If time is (P92) (This is the hourly output.  It will output the next 6 commands.  Time,
            Site, Date, Time, Battery volt, and Sensit Number)
 1: 0        Minutes (Seconds --) into a
 2: 60       Interval (same units as above)
 3: 10       Set Output Flag High (Flag 0)

7:  Resolution (P78)(makes logger able to read numbers above 6999, which all of are site
            numbers are)
 1: 1        High Resolution

8:  Sample (P70)  (Reading site number and putting it out on the hourly data)
 1: 1        Reps
 2: 4        Loc [ site      ]

9:  Resolution (P78)  (logger no longer needs to read large numbers so lowered resolution)
 1: 0        Low Resolution

10:  Real Time (P77)   (recording date and time)
 1: 1110     Year,Day,Hour/Minute (midnight = 0000)

11:  Average (P71)  (averaging the battery voltages for the last hour)
 1: 1        Reps
 2: 3        Loc [ battery   ]

12:  Sample (P70)   (recording the sensit serial number)
 1: 1        Reps
 2: 6        Loc [ snstsnmbr ]

13:  If (X<=>F) (P89)  (if there is a pc reading call subroutine one, which starts the 5
            minute data collection)
```

```
 1: 1        X Loc [ sensit_1  ]
 2: 3        >=
 3: .0001    F
 4: 1        Call Subroutine 1


14:  If (X<=>F) (P89)  (If the timer reaches 5 minutes (3600 seconds) then the logger will
           stop recording 5 minute data)
 1: 5        X Loc [ timer     ]
 2: 3        >=
 3: 3600     F
 4: 21       Set Flag 1 Low

15:  If Flag/Port (P91)  (If the flag is low i.e. no five minute date the program goes to the
           end of the program.)
 1: 21       Do if Flag 1 is Low
 2: 0        Go to end of Program Table

16:  If time is (P92)(if the timer time is less than 5 minutes then carryout the next 4 steps)
 1: 0        Minutes (Seconds --) into a
 2: 5        Interval (same units as above)
 3: 10       Set Output Flag High (Flag 0)

17:  Real Time (P77)  (records time in five minute increments)
 1: 0010     Hour/Minute (midnight = 0000)

18:  Resolution (P78)
 1: 1        High Resolution

19:  Average (P71)  (averages both the pc and the ke)
 1: 2        Reps
 2: 1        Loc [ sensit_1  ]

20:  Totalize (P72)  (totals the pc)
 1: 1        Reps
 2: 1        Loc [ sensit_1  ]



*Table 2 Program
  02: 0.0000    Execution Interval (seconds)

*Table 3 Subroutines

1:  Beginning of Subroutine (P85)
 1: 1        Subroutine 1

2:  Timer (P26)  (resets the timer to zero)
 1: 0        Reset Timer

3:  Do (P86)   (starts recording 5 minute data)
 1: 11       Set Flag 1 High

4:  End (P95)

End Program
```